Zu 13.7.1 Cauchyscher Hauptwert in *Mathematica*

$$\mathcal{J} = \mathrm{PV}\int_a^b \frac{f(x)}{x-c}\,dx = \lim_{\varepsilon\to 0}\left[\int_a^\varepsilon \frac{f(x)}{x-c}\,dx + \int_\varepsilon^b \frac{f(x)}{x-c}\,dx\right]$$

```
$Version
```

```
7.0 for Mac OS X x86 (64-bit) (February 19, 2009)
```

f(x) = 15 (x + 1/3)^2

- **Definitionen**

```
Clear[f, a, b, c]

f[xx_] = 15 (xx + 1 / 3) ^2;
fp[xx_] = D[f[xx], xx];

a = -1 / 2; b = 1; c = 0;
```

- **Graph des Integranden**

```
Plot[{f[x]/(x - c), f[x]}, {x, a, b}, PlotRange → {{a - 0.1`, b + 0.1`}, 30 {-1, 1.5`}},
 PlotStyle → {Dashing[{}], Dashing[{0.02`, 0.03`}]},
 AxesLabel → {"x", "f(x)/(x - c), f(x)"},
 AxesOrigin → {a - 0.1`, 0}, Epilog → {Text["x = c", {0.16`, -28}],
   Text["a", {a, -28}], Text["b", {b, -28}], Text["f(x)", {0.76`, 10}]}]
```



Berechnung des Cauchyschen Hauptwerts in *Mathematica*

- **Das Riemannsche Integral existiert nicht**

Mathematica stellt fest, dass der Integrand so singulär ist, dass das Riemannsche Integal nicht existiert.

```
Integrate[f[x] / (x - c), {x, a, b}]
```

$$\int_{-\frac{1}{2}}^{1} \frac{15\left(\frac{1}{3}+x\right)^2}{x}\,dx$$

Integrate::idiv : Integral of $10 + \frac{5}{3x} + 15x$ does not converge on $\left\{-\frac{1}{2}, 1\right\}$

Analytische Auswertung des Hauptwertintegrals

```
?? Integrate
```

Integrate[f, x] gives the indefinite integral $\int f \, dx$.

Integrate[f, {x, x_{min}, x_{max}}] gives the definite integral $\int_{x_{min}}^{x_{max}} f \, dx$.

Integrate[f, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}, …] gives the multiple integral $\int_{x_{min}}^{x_{max}} dx \int_{y_{min}}^{y_{max}} dy \ldots f$. ≫

```
Attributes[Integrate] = {Protected, ReadProtected}

Options[Integrate] :=
 {Assumptions :→ $Assumptions, GenerateConditions → Automatic, PrincipalValue → False}
```

```
ip = Integrate[f[x] / (x - c), {x, a, b}, PrincipalValue → True]
```

$\frac{5}{24}$ (99 + Log[256])

```
N[ip]
```

21.7802

- **Numerische Berechnung des Hauptwertintegrals**

```
NIntegrate[f[x] / (x - c), {x, a, c, b},
 Method -> {"PrincipalValue", "SingularPointIntegrationRadius" → 1 / 4}]
```

21.7802

▪ Berechnung des Cauchyschen Hauptwerts, wenn f(x) eine stetige Ableitung f'(x) besitzt

Im ersten Integral wird $y = c - x$, im zweiten $y = x - c$ substituiert. Anschließend wird nach y partiell integriert. In den verbleibenden Integranden steht ln(y) und f'.Diese sind also nur mehr schwach singulär an $y = 0$, daher kann man in den Integrationsgrenzen ε gegen Null gehen lassen. Das gibt folgende Formel:

$$\mathcal{J} = \mathrm{PV}\int_a^b \frac{f(x)}{x-c}\,dx = f(b)\ln(b-c) - f(a)\ln(c-a) + \\ + \int_{c-a}^{0} f'(c-y)\ln(y)\,dy - \int_0^{b-c} f'(c+y)\ln(y)\,dy$$

Diese Formel liefert für das obige Integral folgende Ausdrücke:

```
t1 = f[b] Log[b - c]
```

0

```
t2 = - f[a] Log[c - a]
```

$\frac{5\,\mathrm{Log}[2]}{12}$

```
t3 = Integrate[Log[y] fp[c - y], {y, c - a, 0}]
```

$\frac{25}{8} - \frac{3\,\mathrm{Log}[2]}{4} + \mathrm{Log}[4]$

```
t4 = - Integrate[Log[y] fp[c + y], {y, 0, b - c}]
```

$\frac{35}{2}$

```
id = t1 + t2 + t3 + t4 // FullSimplify // Together
```

$\frac{5}{24}\,(99 + \mathrm{Log}[256])$

```
N[id]
```

21.7802

```
ie = Apart[id]
```

$\frac{165}{8} + \frac{5\,\mathrm{Log}[256]}{24}$

- **Sormanns Methode**

- **Graph des Integranden**

```
pl1 = Plot[{f[x]/(x - c)}, {x, a, b}, PlotRange → {{a - 0.1`, b + 0.1`}, 30 {-1, 1.5`}},
   AxesLabel → {"x", "f(x)/(x - c)"}, AxesOrigin → {a - 0.1`, 0},
   Epilog → {Text["x = c", {0.16`, -28}], Text["a", {a, -28}], Text["b", {b, -28}]}];

pl2 = Plot[{f[-x]/(-x - c)}, {x, c, -a},
   PlotRange → {{a - 0.1`, b + 0.1`}, 30 {-1, 1.5`}}, PlotStyle → {Dashing[{0.015`, 0.02`}]},
   AxesLabel → {"x", "f(2c-x)/(c - x)"}, AxesOrigin → {a - 0.1`, 0}];
```

```
Show[pl1, pl2, Ticks → {{{a, ""}, {b, ""}}, None}, ImageSize → 500,
 AxesLabel → {"x", ""}, PlotLabel → "f(x)/(x - c), f(2c-x)/(c - x)",
 Epilog → {Text["c", {0.021`, -28}], Text["a", {a, -28}], Text["b", {b, -28}],
   Text["2c - a", {2 c - a, -28}], Dashing[{0.01`, 0.01`}], Line[{{-a, 40}, {2 c - a, -30}}]}]
```



Die ausgezogene Kurve ist die zu integrierende Funktion g(x) = f(x)/(x - c); deren Singularität bei c ist ein Pol erster Ordnung. Die Funktion g(x) hat bei c einen Vorzeichenwechsel, d.h. bei positivem f(c) geht g(x) links von c gegen -∞, rechts von c gegen +∞. Man sieht, dass sich die nach oben bzw. unten strebenden Äste sich mehr oder minder kompensieren; es bleibt ein endlicher Wert übrig.

```
g[x_] = f[2 c - x] / (c - x) + f[x] / (x - c)
```

$$-\frac{15\left(\frac{1}{3}-x\right)^2}{x}+\frac{15\left(\frac{1}{3}+x\right)^2}{x}$$

```
pp1 = Table[g[x], {x, c, 2 c - a, 0.1}]
```

```
{Indeterminate, 20., 20., 20., 20., 20.}
```

```
ep = 10^(-7);
```

```
pp1 = Table[g[x], {x, c + ep, 2 c - a, 0.1}]
```

```
{20., 20., 20., 20., 20.}
```

```
pp11 = Plot[g[x], {x, c, 2 c - a}];
```

```
pp12 = Plot[g[x], {x, c, 2 c - a}, PlotRange → {20 - ep, 20 + ep}];
```

```
s1 = Integrate[g[x], {x, c, 2 c - a}]
```

```
10
```

$$\texttt{pp2 = Plot}\left[\frac{\texttt{f[x]}}{\texttt{x - c}}, \texttt{\{x, 2 c - a, b\}}\right];$$

```
Show[GraphicsRow[{pp11, pp12, pp2}], ImageSize → 500]
```




```
s2 = Integrate[f[x] / (x - c), {x, 2 c - a, b}]
```

$$\frac{85}{8} + \frac{5\,\text{Log}[2]}{3}$$

```
s1 + s2
```

$$\frac{165}{8} + \frac{5\,\text{Log}[2]}{3}$$

```
N[%]
```

21.7802

```
ie[[1]] + Simplify[ie[[2]]]
```

$$\frac{165}{8} + \frac{5\,\text{Log}[2]}{3}$$